/*
Cidgem Sirin	
Proposal title: Group Empathy as an Explanation for Variation in U.S. Intergroup Reactions to Humanitarian Emergencies Abroad and Refugee Crises


HYPOTHESES

Stated-Hyp1: "We expect African Americans and Latinos should be more likely than Anglos to support military intervention in a civil war, provide financial aid, and offer asylum to the victims" (p. 3)
	
	Test-Hyp1: Support for military aid in a civil war will be higher among African Americans vs. whites.
	
	Test-Hyp2: Support for military aid in a civil war will be higher among Hispanics vs. whites.
	
	Test-Hyp3: Support for financial aid in a civil war will be higher among African Americans vs. whites.
	
	Test-Hyp4: Support for financial aid in a civil war will be higher among Hispanics vs. whites.
	
	Test-Hyp5: Support for asylum in a civil war will be higher among African Americans vs. whites.	

	Test-Hyp6: Support for asylum in a civil war will be higher among Hispanics vs. whites.	
	
	
Stated-Hyp2: "We also expect to observe higher potential for empathy-driven political
action to help the victims among African American and Latino respondents than Anglos" (p. 3)

	Test-Hyp7: Potential for empathy-driven political action will be higher among Blacks vs. whites.
	
	Test-Hyp8: Potential for empathy-driven political action will be higher among Hispanics vs. whites.	

Stated-Hyp3: "We further predict African Americans' and Latinos' support for military intervention and aid will remain significantly higher than Anglos even for cases where the victims do not share common racial/ethnic background with the respondents" (p. 3)

	// Cannot test. Conditions with black and Hispanic victims were removed from survey.
	
	
Stated-Hyp4: "We expect African Americans and Lations to be more supportive of humanitarian action in an Arab country due to their higher group empathy even for outgroups who are generally perceived as threatening in the U.S. society" (p. 3-4)

	Test-Hyp9: Support for humanitarian action in an Arab country will be higher among Blacks vs. whites.
	
	Test-Hyp10: Support for humanitarian action in an Arab country will be higher among Hispanics vs. whites.	

	Test-Hyp11: Higher support for humanitarian action in an Arab country among Blacks vs. whites will be mediated by empathic concern for Arabs.
	
	Test-Hyp12: Higher support for humanitarian action in an Arab country among Hispanics vs. whites will be mediated by empathic concern for Arabs.	

	
	
	
Stated-Hyp5: "The experimental outcomes we will observe are significantly mediated by group empathy -— not perceived threat of national security or ingroup identification" (p. 4)

	// We have already tested effect of group empathy on humanitarian aid.
	// Can't test ingroup identification because there is no measure of construct. 
	
	Test-Hyp13: Higher support for financial aid among African Americans vs. whites will not be mediated by percieved threat of national security.

	Test-Hyp14: Higher support for financial aid among Hispanics vs. whites will not be mediated by percieved threat of national security.
	
	Test-Hyp15: Higher support for asylum among African Americans vs. whites will not be mediated by percieved threat of national security.	

	Test-Hyp16: Higher support for asylum among Hispanics vs. whites will not be mediated by percieved threat of national security.	

	Test-Hyp17: Higher potential for empathy-driven political action among Blacks vs. whites will NOT be mediated by percieved threat of national security.
	
	Test-Hyp18: Higher potential for empathy-driven political action among Hispanics vs. whites will NOT be mediated by percieved threat of national security.	
	
	
********************************************************************************
NOTES:	
- Cannot test stated hypothesis 3. Conditions with black and Hispanic victims 
were removed from survey.
- Questions about empathy-driven political action in the fielded survey differ 
from the questions in the proposal. We have created a summated scale because all 
values range from 0-1.
- For Stated hypothesis 4, we do not know which groups are considered 
threatening. We are assuming they mean Arabs because the question is about Arabs.
- Cannot test part about ingroup identification in stated hypothesis 5 becasue is 
no measure relating to the construct.
- For stated hypothesis 5, we are testing for all outcomes (except humanitarian 
aid support, which is already tested under stated hypothesis 4).

*/

clear all
use "Sirin837.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* victims' racial/ethnic background
	tab XTESS226
	clonevar victimrace= XTESS226
	
* OUTCOME MEASURES

* in case of civil war, military aid, financial aid, asylum
		/*
		V3 [Grid, S Across]

		Considering the violence occurring in [IF XTESS226=1 INSERT: Kornati; IF XTESS226=2 INSERT: Farasan], we now would like to ask you some policy-related questions:

		Rows

		1.	How strongly do you support or oppose using U.S. military forces against the oppressive government of [IF XTESS226=1 INSERT: Kornati; IF XTESS226=2 INSERT: Farasan] in order to help the people experiencing violence? 
		2.	How strongly do you support or oppose the U.S. providing financial aid to the victims of this violent repression?
		3.	How strongly do you support or oppose granting asylum to refugees who attempt to escape from [IF XTESS226=1 INSERT: Kornati; IF XTESS226=2 INSERT: Farasan] to America?
		*/

		forval i=1/3 {
			tab V3_`i', mis
			replace V3_`i'=. if V3_`i'==-1
		}

		gen support_militaryaid = 6-V3_1
		gen support_financialaid = 6-V3_2
		gen support_asylum = 6-V3_3
		

* support for humanitarian action in Syria 

		/*
		Q9 [Grid, S Across]

		As you may know, there has been a violent conflict in Syria between the authoritarian government forces and rebel groups, which has caused thousands of civilian casualties, including many women and children.  How strongly would you support or oppose the United States taking each of the following actions with respect to Syria?

		Rows

	2.	Sending humanitarian aid and supplies to civilians in Syria
		*/
		
		tab Q9_2
		replace Q9_2=. if Q9_2==-1
		gen support_humanitarian = 6- Q9_2 
 
* potential for empathy-driven political action	
		/*
		V4 [S]

		Would you like to receive more information about the situation in [IF XTESS226=1 INSERT: Kornati; IF XTESS226=2 INSERT: Farasan]?  If yes, please click on the box next to the statement "Yes, I would like to receive more information." At the end of the survey, you will be provided a link to the key news feeds and updates about this conflict.  If no, please click "No, I prefer not to receive more information."  In either case, your response will remain anonymous to the researchers who prepared this survey.

		1.	Yes, I would like to receive more information.
		2.	No, I prefer not to receive more information.

		V5 [Q, Range 0.00-1.00, allow two decimal places]

		As mentioned above, for your valuable participation in this survey, you will receive an extra $1.00 (1,000 KP points) credited to your GfK account.  You can donate any part of this $1.00 to a non-governmental organization that works to help the victims of the violent conflict in [IF XTESS226=1 INSERT: Kornati; IF XTESS226=2 INSERT: Farasan].  Please specify the amount of money you would like to donate.  

		You can enter any amount from $0.00 to $1.00. For instance, if you want to donate the whole dollar, simply type "1" in the box below, or if you want to donate 50 cents, enter ".5". If you don't wish to donate any money, simply type "0".

		[Q]$____________

		V6 [S]
		Would you like to sign an electronic petition to be sent to Congress indicating your support for humanitarian aid to help the victims of the violent conflict in [IF XTESS226=1 INSERT: Kornati; IF XTESS226=2 INSERT: Farasan]?  If yes, please click on the "Yes, I would like to sign the petition to help" box below and you will be directed to the petition website at the end of the survey.  If no, please click "No, I prefer not to sign the petition to help."  In either case, your response will remain anonymous to the researchers who prepared this survey.

		1.	Yes, I would like to sign the petition to help.
		2.	No, I prefer not to sign the petition to help.

		*/	

		tab V4, mis
		recode V4 (-1=.) (1=1) (2=0), gen(info)
		
		tab V5, mis
		recode V5 (-1=.), gen(donate)
				
		tab V6, mis
		recode V6 (-1=.) (1=1) (2=0), gen(petition)
		
		gen potentialaction= info+donate+petition
		tab potentialaction
						
		
* MEDIATORS

* group-specific empathy
		forval i=1/4 {
			tab Q3_`i', mis
			replace Q3_`i'=. if Q3_`i'==-1
		}
		
		gen empathic_concern_arabs= 6-Q3_3
		tab empathic_concern_arabs, mis

* percieved threat of national security
		/*
		When it comes to the issue of national security, how concerned are you about each of the following?

		Rows

		1.	The possibility of a terrorist attack on U.S. soil in the near future
		2.	The possibility of you or a friend being the victim of a terrorist attack in the near future
		3.	The possibility of the U.S. getting involved in a high-casualty war overseas that costs the lives of American soldiers
		4.	The possibility of getting involved in a costly war that would hurt the U.S. economy
		*/
		
		// all items are reverse coded. recoding.
		forval i=1/4 {
			tab Q5_`i', mis
			replace Q5_`i'=. if Q5_`i'==-1
		}
		
		gen percievedthreat= Q5_1+Q5_2 +Q5_3+Q5_4
		tab percievedthreat
	
	
* MODERATORS

* respondent's race/ethnicty
		tab PPETHM
		recode PPETHM (1=0 "0 white") (2=1 "1 black") (4=2 "2 Hispanic"), gen(raceeth)
		recode raceeth (1=1) (0=0) (*=.), gen(black)
		recode raceeth (2=1) (0=0) (*=.), gen(hispanic)
		
********************************************************************************

* ANALYSIS


*	Test-Hyp1: Support for military aid in a civil war will be higher among African Americans vs. whites.
	reg support_militaryaid i.raceeth if raceeth!=2
	// do not reject. 0.003
	tess 1.raceeth +, init(Sirin837) bonf(5)
	
*	Test-Hyp2: Support for military aid in a civil war will be higher among Hispanics vs. whites.
	reg support_militaryaid i.raceeth if raceeth!=1
	// do not reject. 0.000
	tess 2.raceeth +, bonf(5)
	
	
*	Test-Hyp3: Support for financial aid in a civil war will be higher among African Americans vs. whites.
	reg support_financialaid i.raceeth if raceeth!=2
	// reject. 0.037. See note about Bonferroni correction.
	tess 1.raceeth +, bonf(5)
	
*	Test-Hyp4: Support for financial aid in a civil war will be higher among Hispanics vs. whites.
	reg support_financialaid i.raceeth if raceeth!=1
	// reject. 0.019. See note about Bonferroni correction.
	tess 2.raceeth +, 	bonf(5)
	
	
*	Test-Hyp5: Support for asylum in a civil war will be higher among African Americans vs. whites.	
	reg support_asylum i.raceeth if raceeth!=2
	// reject. 0.357
	tess 1.raceeth +, bonf(5)
	
	
*	Test-Hyp6: Support for asylum in a civil war will be higher among Hispanics vs. whites.	
	reg support_asylum i.raceeth if raceeth!=1
	// reject. 0.333
	tess 2.raceeth +, 	bonf(5)	
	

*	Test-Hyp7: Potential for empathy-driven political action will be higher among Blacks vs. whites.
	reg potentialaction i.raceeth if raceeth!=2
	// reject. 0.563
	tess 1.raceeth +, bonf(5)
	
	
*	Test-Hyp8: Potential for empathy-driven political action will be higher among Hispanics vs. whites.	
	reg potentialaction i.raceeth if raceeth!=1
	// do not reject. 0.003
	tess 2.raceeth +, bonf(5)
	
	
	
*	Test-Hyp9: Support for humanitarian action in an Arab country will be higher among Blacks vs. whites.
	reg support_humanitarian i.raceeth if raceeth!=2
	// reject. 0.339
	tess 1.raceeth +, bonf(5)
	
	
*	Test-Hyp10: Support for humanitarian action in an Arab country will be higher among Hispanics vs. whites.	
	reg support_humanitarian i.raceeth if raceeth!=1
	// reject. 0.468
	tess 2.raceeth +, 	bonf(5)
	
	
	
*	Test-Hyp11: Higher support for humanitarian action in an Arab country among Blacks vs. whites will be mediated by empathic concern for Arabs.
	est sto clear
	reg support_humanitarian black
	est sto m1

	reg support_humanitarian empathic_concern_arabs black
	est sto m2	
	
	suest m1 m2
	test [m1_mean]black=[m2_mean]black	
	// do not reject. 0.0021
	
	tess, mediation
	
*	Test-Hyp12: Higher support for humanitarian action in an Arab country among Hispanics vs. whites will be mediated by empathic concern for Arabs.	
	est sto clear
	reg support_humanitarian hispanic
	est sto m1

	reg support_humanitarian empathic_concern_arabs hispanic
	est sto m2	
	
	suest m1 m2
	test [m1_mean]hispanic=[m2_mean]hispanic
	// do not reject. 0.0099
	
	tess, mediation	
	
	


*	Test-Hyp13: Higher support for financial aid among African Americans vs. whites will not be mediated by percieved threat of national security.
	est sto clear
	reg support_financialaid black
	est sto m1

	reg support_financialaid percievedthreat black
	est sto m2	
	
	suest m1 m2
	test [m1_mean]black=[m2_mean]black	
	// do not reject. 0.2482
	tess, mediation	

*	Test-Hyp14: Higher support for financial aid among Hispanics vs. whites will not be mediated by percieved threat of national security.
	est sto clear
	reg support_financialaid hispanic
	est sto m1

	reg support_financialaid percievedthreat hispanic
	est sto m2	
	
	suest m1 m2
	test [m1_mean]hispanic=[m2_mean]hispanic
	// do not reject. 0.9096
	tess, mediation	

	
*	Test-Hyp15: Higher support for asylum among African Americans vs. whites will not be mediated by percieved threat of national security.	
	est sto clear
	reg support_asylum black
	est sto m1

	reg support_humanitarian percievedthreat black
	est sto m2	
	
	suest m1 m2
	test [m1_mean]black=[m2_mean]black	
	// do not reject. 0.9415
	tess, mediation	

*	Test-Hyp16: Higher support for asylum among Hispanics vs. whites will not be mediated by percieved threat of national security.		
	est sto clear
	reg support_asylum hispanic
	est sto m1

	reg support_humanitarian percievedthreat hispanic
	est sto m2	
	
	suest m1 m2
	test [m1_mean]hispanic=[m2_mean]hispanic	
	// do not reject. 0.9850
	tess, mediation			
		

*	Test-Hyp17: Higher potential for empathy-driven political action among Blacks vs. whites will not be mediated by percieved threat of national security.
	est sto clear
	reg potentialaction black
	est sto m1

	reg potentialaction percievedthreat black
	est sto m2	
	
	suest m1 m2
	test [m1_mean]black=[m2_mean]black	
	// do not reject. 0.2365
	tess, mediation		
	
*	Test-Hyp18: Higher potential for empathy-driven political action among Hispanics vs. whites will not be mediated by percieved threat of national security.	
	est sto clear
	reg potentialaction hispanic
	est sto m1

	reg potentialaction percievedthreat hispanic
	est sto m2	
	
	suest m1 m2
	test [m1_mean]hispanic=[m2_mean]hispanic
	// do not reject. 0.5868
	tess, mediation	
